

CS412 Exercise sheet 6: solutions

Structuring

1. (a) As the Door machine only SEES the Light machine it is not allowed to include invariant requirements which make reference to Light's variable, switch. Door does not have control over the machine which it SEES and there is **nothing to prevent the light being switched off by some other machine which includes Light**. If this happened, Door's invariant could become false.
- (b) This would produce a consistent machine, but it **wouldn't be the machine required because the light could be switched off but the door open**.
- (c) Could use **USES** which would propagate a proof obligation upwards to be considered later in the development. Or, if nothing else needed to control the light, it could be INCLUDED (or EXTENDED) here.
- (d) It doesn't matter which structuring mechanism is used because the invariant condition will be the same. Neither machine has constraints or properties so the proof obligation is:

$$\begin{aligned}
 & [Init_{Light}; Init_{Door}]Inv_{Door} \\
 & \equiv [switch := off]([door := closed](door \in DSTATE \wedge \\
 & \quad door = open \Rightarrow switch = on)) \\
 & \equiv [switch := off](closed \in DSTATE \wedge (closed = open \Rightarrow switch = on)) \\
 & \equiv closed \in DSTATE \wedge (closed = open \Rightarrow off = on) \\
 & \equiv True \wedge (False \Rightarrow False) \\
 & \equiv True
 \end{aligned}$$

```

2. MACHINE          Fifo(ELEM, cap)
CONSTRAINTS         cap : NAT1
VARIABLES           contents
INVARIANT            contents : seq(ELEM) & size(contents) <= cap
INITIALISATION      contents := <>
OPERATIONS
  input(ee) =
    PRE ee : ELEM & size(contents) < cap
    THEN contents := contents <- ee
    END;
  ee <-- output =
    PRE size(contents) > 0
    THEN ee := first(contents) || contents := tail(contents)
    END
END

```

```

MACHINE          Router
INCLUDES         Fifo(MSG, qmax)
SETS             MSG; DEST; STATUS = {yes, no}

```

```

CONSTANTS      qmax
PROPERTIES      qmax : NAT1
VARIABLES      pending, is_pending, nexthop
INVARIANT       nexthop : MSG --> DEST & pending : MSG & is_pending : STATUS
INITIALISATION  nexthop :: MSG --> DEST || pending :: MSG || is_pending := no
OPERATIONS
  receive(mm) =
    PRE mm:MSG
    THEN IF size(contents) < qmax
      THEN input(mm)
    END
  END;
  retrieve =
    IF size(contents) > 0 & is_pending = no
    THEN pending <-- output || is_pending := yes
    END;
  ndest,msg <-- forward =
    IF is_pending = yes
    THEN msg := pending || ndest := nexthop(pending) || is_pending := no
    END
END

```

The rule for operation PRE P THEN S END (with Router as machine 2 and Fifo as machine 1, C for constraints, B for properties, I for invariant) is:

$$C_1 \wedge C_2 \wedge B_1 \wedge B_2 \wedge I_1 \wedge I_2 \wedge P \Rightarrow [S]I_2$$

In addition, as Fifo has its parameters supplied by the calling machine, we know that $ELEM = MSG$ and $cap = qmax$.

(a) For the receive operation hypotheses (everything to the left of the implication) are:

```

cap : NAT1 & qmax : NAT1 & contents : seq(MSG) & size(contents) <= cap &
nexthop : MSG --> DEST & pending : MSG & is_pending : STATUS & mm:MSG

```

The goal is:

$$[IF \text{ size(contents) } < \text{ qmax } THEN \text{ input(mm) } END]I_2 \\ \equiv \text{ size(contents) } < \text{ qmax } \Rightarrow [\text{ input(mm) }]I_2$$

If we assume the LHS we can then try to prove the RHS.

$$\begin{aligned}
& [\text{ input(mm) }]I_2 \\
& \equiv [PRE \text{ mm : ELEM } \wedge \text{ size(contents) } < \text{ cap } \\
& \quad THEN \text{ contents := contents } \leftarrow \text{ mm } \\
& \quad END]I_2 \\
& \equiv [PRE \text{ mm : MSG } \wedge \text{ size(contents) } < \text{ qmax } \\
& \quad THEN \text{ contents := contents } \leftarrow \text{ mm } \\
& \quad END]I_2 \quad \text{Using known instantiation of parameters.} \\
& \equiv \text{ mm : MSG } \wedge \text{ size(contents) } < \text{ qmax } \wedge [\text{ contents := contents } \leftarrow \text{ mm }]I_2 \quad *
\end{aligned}$$

The first part of this comes directly from the hypothesis list. The second part is the assumption we added when dealing with the implication. Thus we are left with:

$$\begin{aligned}
& [\text{ contents := contents } \leftarrow \text{ mm }]I_2 \\
& \equiv [\text{ contents := contents } \leftarrow \text{ mm }](\text{ nexthop : MSG } \rightarrow \text{ DEST } \wedge \text{ pending : MSG } \wedge \\
& \quad \text{ is_pending : STATUS })
\end{aligned}$$

$$\equiv \text{nexthop} : \text{MSG} \rightarrow \text{DEST} \wedge \text{pending} : \text{MSG} \wedge \text{is_pending} : \text{STATUS})$$

This is simply I_2 and all conjuncts appear in the hypotheses.

- (b) It would be calling the operation from the included machine outside its precondition. Looking at the answer for the previous part, if you haven't got the IF statement, then you wouldn't be adding the extra hypothesis of $\text{size}(\text{contents}) < \text{qmax}$ to the hypothesis list so you wouldn't later be able to discharge the 2nd conjunct from line *.
- (c) For the retrieve operation the hypothesis list is exactly the same except you don't have the final conjunct $\text{mm} : \text{MSG}$ because the operation has no precondition. Our goal is:

$$\begin{aligned} & [\text{IF } \text{size}(\text{contents}) > 0 \wedge \text{is_pending} = \text{no} \\ & \quad \text{THEN } \text{pending} \leftarrow \text{output} \parallel \text{is_pending} := \text{yes} \\ & \quad \text{END}] I_2 \\ & \equiv (\text{size}(\text{contents}) > 0 \wedge \text{is_pending} = \text{no}) \Rightarrow [\text{pending} \leftarrow \text{output} \parallel \text{is_pending} := \text{yes}] I_2 \\ & \quad \wedge \\ & \quad \neg (\text{size}(\text{contents}) > 0 \wedge \text{is_pending} = \text{no}) \Rightarrow I_2 \end{aligned}$$

For the 2nd conjunct, I_2 comes from the hypotheses and so this conjunct is true.

For the 1st conjunct, if we again assume the LHS we're left with the goal:

$$\begin{aligned} & [\text{pending} \leftarrow \text{output} \parallel \text{is_pending} := \text{yes}] I_2 \\ & \equiv [(\text{PRE } \text{size}(\text{contents}) > 0 \\ & \quad \text{THEN } \text{pending} := \text{first}(\text{contents}) \parallel \text{contents} := \text{tail}(\text{contents}) \\ & \quad \text{END}) \parallel \text{is_pending} := \text{yes}] I_2 \\ & \equiv [(\text{PRE } \text{size}(\text{contents}) > 0 \\ & \quad \text{THEN } \text{pending} := \text{first}(\text{contents}) \parallel \text{contents} := \text{tail}(\text{contents}) \parallel \text{is_pending} := \text{yes} \\ & \quad \text{END})] I_2 \\ & \equiv \text{size}(\text{contents}) > 0 \wedge I_2[\text{first}(\text{contents})/\text{pending}, \text{tail}(\text{contents})/\text{contents}, \text{yes}/\text{is_pending}] \end{aligned}$$

The 1st conjunct comes from the hypothesis added when dealing with the implication.

That just leaves:

$$\begin{aligned} & I_2[\text{first}(\text{contents})/\text{pending}, \text{tail}(\text{contents})/\text{contents}, \text{yes}/\text{is_pending}] \\ & \equiv (\text{nexthop} : \text{MSG} \rightarrow \text{DEST} \wedge \text{pending} : \text{MSG} \wedge \text{is_pending} : \text{STATUS})[\text{first}(\text{contents})/\text{pending}, \\ & \quad \text{tail}(\text{contents})/\text{contents}, \\ & \quad \text{yes}/\text{is_pending}] \\ & \equiv \text{nexthop} : \text{MSG} \rightarrow \text{DEST} \wedge \text{first}(\text{contents}) : \text{MSG} \wedge \text{yes} : \text{STATUS} \end{aligned}$$

The first conjunct is a hypothesis. The second conjunct follows since $\text{contents} : \text{seq}(\text{MSG})$ is a hypothesis and the fact that it's nonempty (and therefore $\text{first}(\text{contents})$ is meaningful) is an assumption made when dealing with the implication. The 3rd conjunct is trivially true since yes is one of the defined elements of STATUS .

3. Here is just one possible solution. The main feature from the structuring point of view is the inclusion of 2 separate copies of `Fifo` and the way that these (and their operations and variables) are prefixed. However, there are a few other points worth noting here. Firstly, the requirements are vague about how the decision is made concerning whether to discard a job or pass it on to the next queue. This spec represents it by a nondeterministic choice using the `CHOICE` operator. Secondly, in the operation to move a job from one queue to the next we have to be careful about how we obtain the relevant job from the first queue. This is because the syntax for calling an operation will only allow us to write the call in a place where an operation is expected NOT where an expression is expected. The spec gets round this by referring directly to " $\text{first}(q1.\text{contents})$ " in an `ANY` clause and then using that as the element to add to `q2`. Note that in the `THEN`

section the use of *xx* is merely as a syntactic recepticle for the output from *q1.output*. Things get a lot easier in this respect when we move on to refinement and can use sequencing.

```

MACHINE          JobProcessing

INCLUDES         q1.Fifo(JOBID, qmax1), q2.Fifo(JOBID, qmax2)

SETS            JOBID

CONSTANTS       qmax1, qmax2

PROPERTIES      qmax1 : NAT1 & qmax2 : NAT1

OPERATIONS

/* Operation to receive a new job request onto the first queue
*/
getreq(jj) =
  PRE  jj:JOBID
  THEN IF  size(q1.contents) < qmax1
    THEN q1.input(jj)
  END
  END;

/* Operation to retrieve a queued job id and decide whether to discard it
or move it to the next queue ready for processing.
*/
step1 =
  IF  size(q1.contents) > 0 &  size(q2.contents) < qmax2
  THEN ANY xx WHERE xx : JOBID & xx = first(q1.contents)
    THEN xx <-- q1.output ||
      CHOICE q2.input(xx) OR skip END
    END
  END;

/* Operation to remove job from 2nd queue for processing.
*/
jj <-- step2 =
  IF  size(q2.contents) > 0
  THEN  jj <-- q2.output
  END

END

```